
Условный оператор If.

Полная и неполная форма условного оператора.
Для начала – маленький экскурс в историю. Вспомните окончание 8 класса. Вспомните блок-схемы, которые мы рисовали. У нас тогда появилось такое понятие как ветвление. Ветвление представляло собой некоторый выбор, в зависимости от ситуации,

например:

Ветвление рисовалось как ромбик с вопросом и двумя стрелочками – «да» и «нет».

Главное, что вопрос был именно такой, чтобы ответить на него можно было лишь «да» и «нет». Например, вопрос «какого цвета сигнал светофора?» совершенно тут не подходит.

И поскольку программа – это запись алгоритма на понятном компьютеру языке (чтобы он выполнил!), то в программах тоже можно использовать ветвления.

Для начала, посмотрите на программу:

Program divide;

Var A, B, Result: Real;

Begin

  Write(‘Введите делимое: ’);

  Readln(A);

  Writeln(‘Введите делитель’);

  Readln(B);

  Result:= A / B;

  Writeln(‘Частное: ’ , Result:0:3);

End.

Мы видим, что программа получает от пользователя два числа и сохраняет их в переменные A и B:
  Write(‘Введите делимое: ’);

  Readln(A);

  Writeln(‘Введите делитель’);

  Readln(B);

Затем программа записывает в переменную Result результат деления A на B:
  Result:= A / B;

После чего выводит эту переменную на экран:

  Writeln(‘Частное: ’ , Result:0:3);

Проще говоря, программа находит частное (результат деления) двух введённых чисел.

А теперь задумаемся – всё ли в этой программе хорошо? Вернее, всегда ли. Программа, конечно, откомпилируется и запустится. И даже поделит числа. Но не любые.
Конечно! В случае, если B = 0 программа остановится из-за ошибки. Не умеет компьютер делить на ноль. Да и мы не умеем. Да и не надо этого делать! (
В случае если B = 0, пользователю надо всего лишь вежливо сказать, что такое деление невозможно.
Полная форма условного оператора

Значит, программа должна выбрать сама – можно делить или нет.

Посмотрите на новый текст программы:
Program divide;

Var A, B, Result: Real;

Begin

  Write(‘Введите делимое: ’);

  Readln(A);

  Writeln(‘Введите делитель’);

  Readln(B);

  If (B = 0) Then 

  Begin

    Writeln(‘На ноль делить нельзя!’);

  End

  Else

  Begin

    Result:= A / B;  

    Writeln(‘Частное: ’ , Result:0:3);
  End;

End.

Посмотрите, что добавилось – прочитайте, переведите с английского:
  Если (B равно нулю) то

  Begin

    Writeln(‘На ноль делить нельзя!’);

  End

  в противном случае
  Begin

    Result:= A / B;  

    Writeln(‘Частное: ’ , Result:0:3);
  End;

Эта конструкция и называется условный оператор. И если условие (оно в скобках) выполняется, то программа пишет на экране «На ноль делить нельзя», а если условие не выполняется, то программа уже честно делит и пишет результат.
Ту же программу можно было записать и вот  так:

Program divide;

Var A, B, Result: Real;

Begin

  Write(‘Введите делимое: ’);

  Readln(A);

  Writeln(‘Введите делитель’);

  Readln(B);

  If (B <> 0) Then 

  Begin

    Result:= A / B;  

    Writeln(‘Частное: ’ , Result:0:3);
  End
  Else

  Begin

    Writeln(‘На ноль делить нельзя!’);

  End;
End.

Теперь стоит другие условие – «если B не равно нулю». Знак «не равно» записывается как меньше-больше, т.е. знаки меньше и больше идущие подряд без пробела, вот так: <>
Ну и поскольку мы поменяли условие, то мы переставили местами действия.

Если условие (B не равно нулю) выполняется – программа делит и пишет ответ, если нет - пишет, что делить на ноль нельзя.
Давайте теперь напишем программу, которая будет находить модуль введённого числа.

Напомню, что модуль положительного числа X равен самом числу X, а модуль отрицательного X равен –X. Обозначается модуль вертикальными чертами.
Примеры:

|5| = 5

|-10| = 10

|-17| = 17

|0| = 0

|1| = 1

Программа должна будет получить от пользователя число, а затем сделать выбор, чему будет равен его модуль. Если число положительное или ноль – модуль равен введённому числу. А если число отрицательное – введённому числу но «без минуса». Чтобы отрицательное число «лишить минуса», можно его домножить или поделить на -1.

Заготовка программы:

Program modul_chisla;

Var num, modul: Real;

Begin
  Write(‘Эта программа находит модуль введённого числа’);
  Write(‘Введите число: ’);

  Readln(num);

  Если введённое число больше либо равно нулю, то
  Begin

    Modul:= num;  

  End
  В противном случае
  Begin
    Modul:= num*(-1);  

  End;
  Writeln(‘Модуль равен’ , Modul:0:3);
End.

Теперь переводим условия с человеческого языка в Pascal:

Строчка

  Если введённое число больше либо равно нулю, то
Меняется на

  If (num >= 0) Then
А строчка
  В противном случае
Меняется на

  Else
Вот и всё, готовая программа:

Program modul_chisla;

Var num, modul: Real;

Begin
  Write(‘Эта программа находит модуль введённого числа’);
  Write(‘Введите число: ’);

  Readln(num);

  If (num >= 0) Then
  Begin

    Modul:= num;  

  End
  Else
  Begin
    Modul:= num*(-1);  

  End;
  Writeln(‘Модуль равен’ , Modul:0:3);
End.

Неполная форма условного оператора
Иногда нам нудно что-то делать, только если условие выполняется. Блок-схема имеет следующий вид:

Такой вариант условного оператора тоже существует.

Изменим нашу программу, находившую модуль. Теперь она будет превращать введённое число в его модуль, причём в той же самой переменной num
Т.е. что-то делаем с числом в переменной num только если оно отрицательное – тогда переменную num надо домножить на -1, в противоположном случае ничего с ним делать не надо:

Program modul_chisla;

Var num: Real;

Begin
  Write(‘Эта программа находит модуль введённого числа’);
  Write(‘Введите число: ’);

  Readln(num);

  If (num < 0) Then
  Begin
    num:= num*(-1);  

  End;
  Writeln(‘Модуль равен’ , num:0:3);
End.

Упражнение:
1. Напишите программу, пользователь которой будет вводить свой возраст в годах (число), а программа будет сообщать, сколько это в месяцах. Но если человек введёт ноль или отрицательное число – надо ему сообщить, что такой возраст не существует и перевод в месяцы не делать.
2. Напишите программу, которая будет получать два числа и сообщать, равны они или нет.

